Stata syntax for internal consistency estimation in the four practical scenarios described in Viladrich, C., Angulo-Brunet, A., & Doval, E. (2017). A journey around alpha and omega to estimate internal consistency reliability. Annals of Psychology, 33(3), 755-782. doi: 10.6018/analesps.33.3.268401

See http://ddd.uab.cat/record/173917 for datasets and Table 1 and Table 2 in the paper for selected output obtained using R.

See http://ddd.uab.cat/record/205870 for datasets without variable names, as are use din the present syntax. Output using Stata could difer from Table 1 and Table 2 due to diferences in computational algorithms.

Table of contents



```
*Starting Stata
*Defining and checking the working directory
cd "C:\workingdirectory"
pwd

*Installing and checking packages needed to perform the analyses
*Don't run if already installed!
ado
ssc install tab_chi
net install polychoric, from(http://staskolenikov.net/stata)
ssc install relicoef
ado


*Case 1: essentially tau-equivalent measures
*Reading data
infile y1 y2 y3 y4 y5 y6 using "Case1_noNames.txt", clear

*Phase 1
*Response percentages
tabm y1-y6, row
*Other univariate statistics
tabstat y1-y6, statistics( count min max mean sd skewness kurtosis ) noseparator columns(statistics)
*Pearson correlations
correlate y1-y6

*Phase 2
*Specification, estimation and fit of the essentially tau-equivalent measurement model
sem (F1tau@la -> y1-y6),var(F1tau@1)
estat gof, stats(all)
sem,standardized

*Specification, estimation and fit of the congeneric measurement model
sem(F1cong -> y1-y6), var(F1cong@1)
estat gof, stats(all)
sem,standardized
```

```
*Phase 3
*Point estimation of coefficient alpha
alpha y1-y6
*Point estimation of coefficient omega for essentially tau-equivalent measures
sem (F1tau@la -> y1-y6),var(F1tau@1)
relicoef
*Interval estimation of coefficient alpha
qui: alpha y1-y6
scalar n_row=_N
scalar CA=r(alpha)
scalar n_col=c(k)
scalar CA_l=1-[(1-CA)*invFtail(n_row-1,  (n_row-1)*(n_col-1), .025)]
scalar CA_u=1-[(1-CA)*invFtail(n_row-1, (n_row-1)*(n_col-1),1-.025)]
display as text "Cronbach's alpha= " %04.3f CA "; IC95% = [ " %04.3f CA_l " : " %04.3f CA_u " ]"
*Interval estimation of coefficient omega not included


*Case 2: congeneric measures
*Reading data
infile y1 y2 y3 y4 y5 y6 using "Case2_noNames.txt", clear

*Phase 1
*Response percentages
tabm y1-y6, row
*Other univariate statistics
tabstat y1-y6, statistics( count min max mean sd skewness kurtosis ) noseparator columns(statistics)
*Pearson correlations
correlate y1-y6

*Phase 2
*Specification, estimation and fit of the essentially tau-equivalent measurement model
sem (F1tau@la -> y1-y6),var(F1tau@1)
estat gof, stats(all)
sem,standardized

*Specification, estimation and fit of the congeneric measurement model
sem(F1cong -> y1-y6), var(F1cong@1)
estat gof, stats(all)
sem,standardized

*Phase 3
*Point estimation of coefficient alpha
alpha y1-y6
*Point estimation of coefficient omega for congeneric measures
sem(F1cong -> y1-y6), var(F1cong@1)
relicoef
*Interval estimation of coefficient alpha
qui: alpha y1-y6
scalar n_row=_N
scalar CA=r(alpha)
scalar n_col=c(k)
scalar CA_l=1-[(1-CA)*invFtail(n_row-1,  (n_row-1)*(n_col-1), .025)]
scalar CA_u=1-[(1-CA)*invFtail(n_row-1, (n_row-1)*(n_col-1),1-.025)]
display as text "Cronbach's alpha= " %04.3f CA "; IC95% = [ " %04.3f CA_l " : " %04.3f CA_u " ]"
*Interval estimation of coefficient omega not included


*Case3: measures with correlated errors
```

```
*Reading data
infile y1 y2 y3 y4 y5 y6 using "Case3_noNames.txt", clear

*Phase 1
*Response percentages
tabm y1-y6, row
*Other univariate statistics
tabstat y1-y6, statistics( count min max mean sd skewness kurtosis ) noseparator columns(statistics)
*Pearson correlations
correlate y1-y6

*Phase 2
*Specification, estimation and fit of the essentially tau-equivalent measurement model
sem (F1tau@la -> y1-y6), var(F1tau@1)
estat gof, stats(all)
sem,standardized

*Specification, estimation and fit of the congeneric measurement model
sem(F1cong -> y1-y6), var(F1cong@1)
estat gof, stats(all)
sem,standardized

*Specification, estimation and fit of the measurement model with correlated errors
sem (F1err_cor -> y1-y6), var(F1err_cor@1) cov(e.y4*e.y5) cov(e.y4*e.y6) cov(e.y5*e.y6) nolog
estat gof, stats(all)
sem,standardized

*Phase 3
*Coefficient alpha not included
*Point estimation of coefficient omega for measures with correlated errors
sem (F1err_cor -> y1-y6), var(F1err_cor@1) cov(e.y4*e.y5) cov(e.y4*e.y6) cov(e.y5*e.y6)
relicoef
*Interval estimation of coefficient omega not included

*Case 4: ordered categorical data
*Reading data
infile y1 y2 y3 y4 y5 y6 using "Case4_noNames.txt", clear

*Phase 1
*Response percentages
tabm y1-y6, row
*Other univariate statistics
tabstat y1-y6, statistics( count min max mean sd skewness kurtosis ) noseparator columns(statistics)
*Polychoric correlations
polychoric y1-y6

*Phase 2 using ML estimator and ordinal probit link
*Specification, estimation and fit of the essentially tau-equivalent measurement model
gsem (F1tau@la -> y1-y6),var(F1tau@1) oprobit
estat ic
*Save data for model comparison
estimates store tau

*Specification, estimation and fit of the congeneric measurement model
gsem (F1cong -> y1-y6), var(F1cong@1) oprobit
estat ic
*Comparison with previous model
lrtest tau
```

```
*Phase 3
*Point estimation of coefficient alpha ordinal
polychoric y1-y6
matrix define C = r(R)
factormat C, n(600) factors(1)
tempname L Psi
matrix define `L' = e(L)
matrix define `Psi' = e(Psi)
local p = rowsof(`L')
tempname f f2 u2
scalar define `f' = 0
scalar define `f2' = 0
scalar define `u2' = 0
forvalues i = 1/`p' {
scalar define `f' = `f' + `L'[`i', 1]
scalar define `f2' = `f2' + `L'[`i', 1] * `L'[`i', 1]
scalar define `u2' = `u2' + `Psi'[1, `i']
}
scalar define `f' = `f' / `p'
scalar define `f2' = `f2' / `p'
scalar define `u2' = `u2' / `p'
tempname pf2
scalar define `pf2' = `p' * `f' * `f'
scalar define alphao = `p' / (`p' - 1) * (`pf2' - `f2') / (`pf2' + `u2')
display in smcl as text "Ordinal alpha = " as result %06.4f alphao

*Point estimation of coefficient omega ordinal
*run the next syntax from a .do file
local thevars y1 y2 y3 y4 y5 y6
polychoric `thevars'
mat polychR = r(R)
forvalues i=1/`: word count `thevars'' {
  forvalues j=1/`i' {
    local setcor `setcor' `=polychR[`i',`j']'
  }
  if `i' < `: word count `thevars'' local setcor `setcor' \
 }
 local N = _N
 clear
 ssd init `thevars'
 ssd set obs `N'
 ssd set cor `setcor'
sem (F1cong -> y1-y6), var(F1cong@1)
relicoef

*Point estimation of coefficient omega categorical not included

*Interval estimation of reliability coefficients not included
```